
0.  INTRODUCTION
This program written in FORTRAN77 calculates the viscous component in roll damping.  The method used as the basis is a component discrete type roll damping prediction method developed in Osaka Prefecture University.  Here, simplified description of the component discrete type roll damping prediction method and description of an input part are shown.  In addition, the notation currently used in the program is indicated with a program.
1.  CHARACTERISTICS OF ROLL DAMPING
In the ship motion in waves, most of hydrodynamic forces acting on hull can be calculated using a potential theory.  However, a roll damping is affected the viscous effect significantly.  Therefore, a calculated result by a potential theory overestimates the roll amplitude in resonance and it is not practical.  So, it is common for a calculation of roll damping to use measured values or estimation method in consideration of a viscosity.
The roll damping prediction method used here is a component discrete type prediction method which was developed by Ikeda et al and which was developed for general-cargo type of ships (round bilge vessel).  The roll damping is consisted with five components which are friction component, wave making component, eddy making component, lift component and bilge-keel component.  It has been confirmed that estimated results for general-cargo type of ships have adequate accuracy.  In the following section, each component is briefly explained with references.
1-1)  Frictional Component
Ref 1)
Hiroshi Kato: On the Frictional Resistance to the Rolling of Ships, Journal of the Society of Naval Archiect of Japan, No.102, Nov. 1957, pp.115-122 (in Japanese).
Ref 2)
Shin Tamiya and Takashi Komura: Topics on Ship Rolling Characteristics with Advance Speed, Journal of the Society of Naval Architects of Japan, No.132, Oct. 1972, pp.159-168 (in Japanese).



Frictional component takes 8 to 10% of a total roll damping for about 2m in overall-length model ship.  This component has Reynolds effect (scale effects), and the rate decreases in proportion to ship size and it takes about 1 to 3% for real scaled ships.  Other components of a roll damping do not have a scale effect.  Therefore, even if the scale of a ship is changed, the non-dimensional damping coefficient can be used as the same value.  For estimation of this component, the following equation which is the modified Kato’s equation at Fn=0 by Tamiya’s advanced-speed modification is used.

In the equation, Cf (; the equivalent flat plate frictional coefficient), rf (the equivalent radius), Sf (the surface area) are denote the following equations.

Where ν is the coefficient of kinematic viscosity, T is roll period, 

 is distance from the still water surface to the center of gravity (down ward +).
1-2)  Wave Making Component
Ref 3)
Yoshiho Ikeda, Yoji Himeno and Norio Tanaka: Component of Roll Damping of Ship at Forward Speed, Journal of the Society of Naval Architects of Japan, No.143, May 1978, pp.121-133 (in Japanese).   (+ in English)
This component takes about 5 to 30% for general-cargo type ships.  However, the component may account big rate for a kind of ships with shallow draught and wide section.  The component at Fn=0 can be calculated by a linear potential theory based on small wave height.  The effects of forward speed the component is taken into consideration using the following modification equation.

In addition, in recent years, the component with the effects of forward speed is computable with an exact potential calculation.
1-3)  Eddy Making Component by Naked Hull
Ref 4)
Yoshiho Ikeda, Yoji Himeno and Norio Tanaka: On Eddy Making Component of Roll Damping Force on Naked Hull, Journal of the Society of Naval Archtects of Japan, No.142, Nov. 1977, pp.59-69.
Ref 3)
Yoshiho Ikeda, Yoji Himeno and Norio Tanaka: Component of Roll Damping of Ship at Forward Speed, Journal of the Society of Naval Architects of Japan, No.143, May 1978, pp.121-133 (in Japanese).
Ref 5)
Yoshiho Ikeda: Roll Damping, Txst of 1st symposium on stability of ships, the Society of Naval Architecs of Japan, Dec. 1984, pp.241-250.
Ref 6)
Y. Ikeda, T. Fujiwara, T. Katayama: Roll Damping of a Sharp-Cornered Barge and Roll Control by a New-Type Stabilizer, Proc. of 3rd ISOPE Vol.III, June 1993, pp.643-639.
Ref
Toru Katayama, Naofumi Yoshida, Yusuke Yamamoto, Masashi Kashiwagi, Yoshiho Ikeda: Effects of Height of Center of Rolling and Shallow Draught on Eddy Making Component of Roll Damping - Correction and Modification of Eddy Making Component in Ikeda’s method -, Proceedings of JASNAOE, No.33, Nov.2021, pp.
It is the damping component caused by vortex shedding as well as the bilge-keel component.  The local two-dimensional vortex shedding from bilge and bottom around bow and stern, where pressure gradient changes rapidly, is the main cause.  However, in this case, since KC number (Keulegan-Carpenter number) is very small, vortex shedding is not large.  Here, the pressure distribution caused on the hull surface by the vortex shedding is approximated by the first-order equation, and that pressure value is determined from the measured results of the two-dimensional models which have various shapes of cross-section.  The estimate equation for a two-dimensional section is expressed with the following equation. 

γ in the previous equation is the ratio of the maximum current velocity and mean velocity on the hull surface.  This can be obtained by the calculation for the Lewes form section.  Since the roll amplitude is contained in the equation of BE, the component is nonlinear component for roll amplitude.  The eddy making component decreases quickly with the increment of forward speed.  The forward speed effect is approximately expressed with the following equation. 

  where, 

1-4)  Lift Component by Naked Hull
Ref 3)
Yoshiho Ikeda, Yoji Himeno and Norio Tanaka: Component of Roll Damping of Ship at Forward Speed, Journal of the Society of Naval Architects of Japan, No.143, May 1978, pp.121-133 (in Japanese).
Under quasi-static assumption, the lift component is expressed with the following equation.

where 

The lift gradient kn to calculate maneuverability is used and it is expressed the following equation.

It should be noted that the above mention indicate only the lift component by the naked hull.  If the ship has large bilge keels or skeg and so on, these appendages may cause same component.  In this case, these components can be considered with same manner (Ref 9).
1-5)  Bilge Keel Component
Ref 7)
Yoshiho Ikeda, Yoji Himeno and Norio Tanaka: On Roll Damping Force of Ship – Effects of Friction of Hull and Normal Force of Bilge Keels -, Journal of Kansai Society of Naval Architecture, Japan, Vol.161, June  1976, pp.41-49.
Ref 8)
Yoshiho Ikeda, Kiyoshi Komatsu, Yoji Himeno and Norio Tanaka: On Roll Damping Forceof Ship – Effects of Hull Surface Pressure Created by Bilge-keels -, Journal of Kansai Society of Naval Architecture, Japan, Vol.165, June 1977, pp.31-40.
Ref 9)
Yoshiho Ikeda, Toru Katayama, Yuji Hasegawa and Masayuki Segawa: Roll Damping of High Speed Slender Vessels, Journal of Kansai Society of Naval Archtects, Japan, Vol.222, Sept. 1994, pp.73-81.
The bilge keel component is divided into normal pressure component on bilge keel and the hull surface pressure component.  The first is based on the pressure acting on the bilge keel itself, and the second is based on the pressure fluctuation caused by the vortex shedding from the bilge keel on the hull surface.  The normal pressure component on the bilge keel is given by the following equation.

Where f is the modification coefficient about a bilge radius and it is expressed by the following equation.

Where, r is the distance from the center of rolling to the bilge-keel, bBK and lBK are the width and length of a bilge keel, 0 is roll amplitude.  The hull surface pressure component is calculated by integrating with the pressure distribution which a bilge keel makes along the hull surface.  In addition, two kinds of algorithms can be used in this program.  (1) is the simplified estimation method.  (2) is the method that the attachment condition of a bilge keel can be more correctly taken into consideration.  (2) should be used for the fine-ship type.
2.  INPUT DATA & OUTPUT DATA
The input file name of the following program is "OSM.RDP", an output file name is "OSMB44.CSV", and both files are text files.  The input file is written the calculated condition those are hull data, forward speed, roll amplitude, wave period and wave direction (roll period).  On the other hand, for the output file, the conditions of calculation, the calculated roll damping coefficient and its components are written.  In order to know the detail of the input file, please refer to the "SA_RDPM_new_inputfile_format.xls".

3.  ROLL CALCULATION
Ref 10)
Fukuda, J., Nagamoto, R., Konuma, M., and Takahashi, M.: "Theoretical Calculations on the Motions, Hull Surface Pressures and Transverse Strength of a Ship in Waves," Journal of the Society of Naval Architects of Japan, Japan, Vol. 129 (1971) (in Japanese).
Ref 11)
Yamanouchi, Y.: An Application of Multi-Spectrum Analysis to Ship Responses and Treatment of Nonlinear Response," Journal of the Society of Naval Architects of Japan, Vol. 125 (1969).
Ref 12)
Kaplan, P.: "Nonlinear Theory of Ship Roll Motion in a Random Seaway,"Webb !nst. of Naval Arch., Lec. Notes (1966).
Since the viscous component of a roll damping changes according to the change in roll amplitudes, total roll damping is nonlinear.  

However, since this nonlinearity appears only for roll amplitude strongly, it can use the technique of equivalent linearization for energy in a roll motion period.  
In the case of regular sea, the linearization roll damping B44 (for quadratic term) is obtained as following

The roll damping coefficient B44 is a function of roll amplitude 0, and it becomes difficult to solve roll motion equation.  The iteration method for roll amplitude is used as a method for a frequency domain roll calculation. (Ref.10)
In the case of irregular seas, to predict the ship response in short-term irregular seas, the energy-spectrum analysis based on the principle of linear superposition has been applied.  Let the energy spectrum of the incoming long-crested waves be denoted by Sw(ω).  Then we can obtain the energy spectrum of the ship response S(ω) in the form

The frequency ω of the waves can be transformed into the encounter frequency ωe of the ship.  The response amplitude operator A(ω) represents the ship response per unit magnitude of input wave.  In particular, it becomes 0 > kha (k : the wave number) in the case of roll response.  The equation system is nonlinear when A(ω) depends on the amplitude of the wave and thus on the spectrum Sw(ω).  Several attempts have been made to treat this nonlinearity.  One of method is to take the A(ω) value is the regular perturbation technique to the nonlinear terms in the equation of motion.  The concept of the equivalent linear damping has been applied to this case:

This has been derived from the assumption that the difference between linear and nonlinear damping can, in the sense of least-squares method, be minimized during the roll motion, which is assumed to be a Gaussian process. 

under assumption 

This treatment also requires an iteration method of solution, in which the variance 

and should be reevaluated in each iteration stage.(Ref.11)  As a more simplified way to avoid iteration calculation, it is proposed that the variance be replaced by the quantity

Here, ωn is the roll natural frequency.(Ref.12)  These ways can be also applied to predict transitional time domain roll motion using convolution integration method.  However, it is should be noted that these models are proposed from the statistical point of view and are not necessarily consistent with the physical point of view.  Near future, it is hoped that a new method is proposed.
4.  COPYRIGHT, DISCLAIMER, REDISTRIBUTION OF PROGRAM
  This software is freeware.  It can be used free of charge.  The program author's Toru Katayama and Yoshiho Ikeda hold copyright.  This software is not guaranteed.  No matter what trouble may occur by using this program, the authors cannot take responsibility.  Redistribution by the non-profit of this program can be freely carried out, as long as no change is carried out.  When you wish the inclusion to books, a magazine, etc., and redistribution for the profit purpose, please contact authors.  When you wish the inclusion of calculated results to journal, proceedings, etc., for research, please include "CAD system for ships, Yoshiho Ikeda and Toru Katayama, Ship and Port Editing Room (2001)", as a citation.
Yoshiho IKEDA  ikeda@marine.osakafu-u.ac.jp
Toru KATAYAMA  katayama@marine.osakafu-u.ac.jp
4.  PROGRAM LIST AND DESCRIPTION OF NOTATIONS
C**********************************************************************

C*  Estimation of Roll Damping for conventional cargo ship & barge              *

C*                                                                    *

C*  SA_RDPM.BAS                                                    *

C*                                      Osaka Prefecture University       *

C**********************************************************************

C

      PROGRAM MAIN

      IMPLICIT DOUBLE PRECISION (A-H,O-Z) 

      DOUBLE PRECISION NUE,L,KAI,L0,L1,L2,NABLA,KW,KAI1,LTHETA,MGA,KN

     &    ,LO,LR,KG,LB1,LB2,LB3,LB2S0,M1,M2,M3,M4,M5,M6,M7,M8

      INTEGER BWOSM,TYPE_BK,RDP,CSDMIN,SSBK1,SSBK2

C     Max. number of square stations: SC, Max. number of points in a square station: PT
C     Max. number of wave lengths: NNRA, Max. number of Fn: NNFN
C     Max. number of wave directions: NNKAI, Max. number of roll amplitudes: NNRD

      PARAMETER SC=200, PT=200, NNRA=70, NNFN=20, NNKAI=35, NNRD=30

C

      DIMENSION  CSD(SC), NUMPT(SC), HT(0:SC,0:PT), HB(0:SC,0:PT)

      DIMENSION  THETA1(NNRD), RAM(NNRA), FUN(NNFN), KAI(NNKAI)

      DIMENSION  XX(SC,PT*2), YY(SC,PT*2)

      DIMENSION  X(SC), HO(SC), SIG(SC), BX(SC), DX(SC)

      DIMENSION  PERI(NNRA,NNFN,NNKAI), BE2(SC)

      DIMENSION  RMAX1(2), V(2), CR(SC), CR1(SC)

      DIMENSION  X1(30), YI(30), XBK(0:SC), RATIO(SC)

      DIMENSION  X11(SC), Y11(SC), WX(SC), WY(SC)

      DIMENSION  BKT(SC), BKB(SC), NUBK(SC), BBKHAT(SC)

      DIMENSION  BBYD(SC), SGM(SC), BB0(SC), TT0(SC), WLHB(SC), WLAR(SC)

      DIMENSION  GLEN(SC)

C

C

C

C******

C    if BWOSM=0, wave making comp is not calculated.

C    if BWOSM=1, wave making comp is calculated by simplified calculation.
      BWOSM=0

C     Coefficient of kinematic viscosity: NUE
      NUE=1.14E-06

C     Circular constant : PAI
      PAI=3.14159

C     Gravitational acceleration: G, Fluid density: RO
      G=9.8

      RO=102.0

C

C     loading the data file which is written hull and bilge keel data and sea conditions: OSM.RDP
       OPEN(1, FILE='OSM.RDP')

C****** Hull form data
C     Lpp[m]：L、Breadth[m]：B、Draft[m]：D
       READ(1,*)L, B, D

C     number of cross section in offset data: MUMCS, the number indicating mid-ship section: IMID
       READ(1,*)NUMCS, IMID

       DO I=1, NUMCS

C     SS number：CSD(I)
         READ(1,*) CSD(I)

C     number of datum in a square section: NUMPT(I)
         READ(1,*) NUMPT(I)

         DO J=1 , NUMPT(I)

C     height from Base Line[m]: HT(I,J), breadth from Center Line[m]: HB(I,J) (please input from BL & CL）
           READ(1,*) HT(I,J), HB(I,J)

         END DO

       END DO

C****** information of bilge keel
C     breadth of bilge keel [m]: BBK, bilge keel aft-end ss number: XBK1, bilge keel forward-end ss number: XBK2
       READ(1,*) BBK, XBK1, XBK2

C     Selection of calculation methods for bilge keel component (TYPE_BK= 1: simplified, 2: integration alone hull form)
       READ(1,*) TYPE_BK

C     Attention) if TYPE_BK=2 them, the datum from “IF ( TYPE_BK.NE.1) then” to “END IF” are loaded.
       IF (TYPE_BK.NE.1) then

C     The aft-end and the forward-end of the section data number (1 to MUMCS) in the program where the bilge keel exists on hull.: SSBK1 & SSBK2 ( 1 to MUMCS)
         READ(1,*) SSBK1, SSBK2

         DO I=SSBK1, SSBK2
C     The attached transverse position of the bilge-keel on the section from CL [m]：BKB(I)
           READ(1,*) BKB(I)

         END DO

       END IF

C*****Selection of calculation methods for eddy making component (RDP=1: conventional hull, 2:barge)
       READ(1,*) RDP

C*****OGD=(Draft-KG)/Draft

       READ(1,*) OGD

C*****Number of roll amplitudes for calculation: NTHETA
       READ(1,*) NTHETA

       DO I=1 , NTHETA

C*****Roll amplitude [degree]：THETA1(I)
         READ(1,*) THETA1(I)

       END DO

C*****Number of Froude numbers (Fn) forculculation: NFN
       READ(1,*) NFN

       DO I=1 , NFN

C*****Froude number (Fn)：FUN(I)
         READ(1,*) FUN(I)

       END DO

C*****Number of wavelength/Lpp for calculation: NRA
       READ(1,*) NRA

       DO I=1, NRA

C*****wavelength/Lpp: RAM(I)
         READ(1,*) RAM(I)

       END DO

C*****Number of wave directions for calculation: NKAI
       READ(1,*) NKAI

       DO I=1, NKAI

C*****wave direction(degree),  0=following sea, 180=head sea: KAI(I)
         READ(1,*) KAI(I)

       END DO

       CLOSE(1)

C===== 

C

C=====When a bilge-keel component is calculated by the integration method, OFFSET DATA is extended to right and left.
        DO I=1 , NUMCS

          HT(I,NUMPT(I)+1)=999 

          HB(I,NUMPT(I)+1)=HB(I,NUMPT(I))

        END DO

        DO I=1 , NUMCS

          DO J=1 , NUMPT(I)

            XX(I,NUMPT(I)-J+1)=HB(I,J)

            XX(I,NUMPT(I)+J-1)=-HB(I,J)

            YY(I,NUMPT(I)-J+1)=HT(I,J)

            YY(I,NUMPT(I)+J-1)= HT(I,J)

          END DO

        END DO

C

C====== calculation of hull form data =====

C=====half breadth：WLHB(I)、water line area：WLAR(I)、girth length：GLEN(I)
        WLHT=D

        DO I=1 , NUMCS

          SUM=0.0

 10    CONTINUE

          IF( HT(I,1).GT.WLHT)THEN

            WLHB(I)=0.

            WLAR(I)=0.

            GLEN(I)=0.

            GOTO 10

          END IF

          IF( HT(I,1).EQ.WLHT)THEN

            WLHB(I)=HB(I,1)

            WLAR(I)=0.

            GLEN(I)=HB(I,1)

          END IF

          DO J=1 , NUMPT(I)

 11    CONTINUE

            IF( (HT(I,J).LT.WLHT) .AND. (WLHT.LE.HT(I,J+1)))THEN

              GOTO 12

            END IF

            SUM=SUM+(HT(I,J+1)-HT(I,J))*(HB(I,J+1)+HB(I,J))/2

            GOTO 14

C      ====== calculation on each water line

 12    CONTINUE

          AA=(HB(I,J+1)-HB(I,J))/(HT(I,J+1)-HT(I,J))

          BB=(-HB(I,J+1)*HT(I,J)+HB(I,J)*HT(I,J+1))/(HT(I,J+1)-HT(I,J))

            WLHB(I)=AA*WLHT+BB

            WLAR(I)=SUM+(WLHT-HT(I,J))*(WLHB(I)+HB(I,J))/2

            GOTO 15

 14    CONTINUE

          END DO

 15    CONTINUE

        END DO

C      ===== 

C     H0: BBYD(I), σ: SGM(I), draft of a section/mean draft: TT0(I)
C     water line breadth of a section/Breadth: BB0(I)

       DO I=1, NUMCS

         IF (HT(1,1).LT.D) THEN

           CSDMIN=1

         END IF

         IF ((HT(I-1,1).GT.D).AND.(HT(I,1).GE.D)) THEN

           GOTO21

         END IF

         IF ((HT(I-1,1).GE.D).AND.(HT(I,1).LT.D)) THEN

           CSDMIN=I

         END IF

         IF ((HT(I,1).GE.D).AND.(HT(I-1,1).LT.D)) THEN

           NUMH0=I-1

           GOTO 22

         END IF

         HB2=HB(I,1)

         DO J=2, NUMPT(I)

           IF (HT(I,J).GT.WLHT) THEN

             GOTO 18

           END IF

           IF (HB(I,J).GT.HB2) THEN

             HB2=HB(I,J)

           END IF

         END DO

 18      CONTINUE

         IF (WLHB(I).GT.HB2) THEN

           HB2=WLHB(I)

         END IF

         IF (WLHB(I).EQ.0) THEN

           BBYD(I)=HB2/(WLHT-HT(I,1))

           GOTO 19

         END IF

         BBYD(I)=WLHB(I)/(WLHT-HT(I,1))

 19      CONTINUE

         IF (WLHB(I).EQ.0) THEN

           SGM(I)=WLAR(I)/(HB2*(WLHT-HT(I,1)))

           GOTO 20

         END IF

           SGM(I)=WLAR(I)/(WLHB(I)*(WLHT-HT(I,1)))

 20      CONTINUE

         TT0(I)=(WLHT-HT(I,1))/D

         IF (TT0(I).LT.0) THEN

           TT0(I)=0

         END IF

         IF (WLHB(I).EQ.0) THEN

           BB0(I)=HB2/B*2

           GOTO 21

         END IF

         BB0(I)=WLHB(I)/B*2

 21      CONTINUE

       END DO

       NUMH0=NUMCS

 22    CONTINUE

C

       N=NUMCS-CSDMIN+1

       DO I=1, N

         X(I)  =CSD(CSDMIN-1+I)

         HO(I) =BBYD(CSDMIN-1+I)

         SIG(I)=SGM(CSDMIN-1+I)

         BX(I) =BB0(CSDMIN-1+I)*B

         DX(I) =TT0(CSDMIN-1+I)*D

       END DO

C

C====== calculation of displacemental volume
C      displacemental volume [m3]：NABLA

       ODD=0.0

       if (mod(NUMCS,2).LT.0.5) then

         ODD=1.0

       end if

       DO J=1, NUMCS-2-ODD , 2

         Q0=CSD(J)

         Q1=CSD(J+1)

         Q2=CSD(J+2)

         QQ=(Q0+Q2)/2

         HH=(Q2-Q0)/2*L/10/3

         L0=4*(QQ-Q1)*(QQ-Q2)/(Q0-Q1)/(Q0-Q2)+1

         L1=4*(QQ-Q0)*(QQ-Q2)/(Q1-Q0)/(Q1-Q2)

         L2=4*(QQ-Q0)*(QQ-Q1)/(Q2-Q0)/(Q2-Q1)+1

         SUM3=SUM3+(L0*WLAR(J)+L1*WLAR(J+1)+L2*WLAR(J+2))*HH

       END DO

       IF (ODD.LT.0.5) THEN

         GOTO 1404

       END IF

       J=NUMCS-1

       HH=(CSD(J+1)-CSD(J))/2*L/10

       SUM3=SUM3+(WLAR(J)+WLAR(J+1))*HH

 1404  NABLA=2*SUM3

C

C====== calculation of block coefficient Cb: CB, midship coefficient CM: CM
       CB=NABLA/L/B/D

       CM=2*WLAR(IMID)/B/D

C

C====== Calculation of the encounter period for calculating roll damping
C      wave length λ [m]: RAM(I), Fn: FUN(J), wave direction χ [deg]: KAI(K)
C      encounter frequency [rad/sec]: MGA, encounter period [sec]: PERI(I,J,K)
       DO K=1, NKAI

         DO J=1, NFN

           DO I=1, NRA

             OMG=SQRT(2*PAI*G/RAM(I)/L)

             KW=2*PAI/RAM(I)/L

             V2=FUN(J)*SQRT(L*G)

             KAI1=KAI(K)*PAI/180

             MGA=OMG-KW*V2*COS(KAI1)

             MGA=ABS(MGA)

             PERI(I,J,K)=2*PAI/MGA

           END DO

         END DO

       END DO

C

C=====The attachment height of a bilge keel is calculated using a bilge-keel transverse direction attachment position and hull form data. [m]: BKT(I)

C=====In the data in a section, it is verified whether the bilge keel is attached between the data of what position.
       IF (TYPE_BK.NE.1)THEN

         DO I=SSBK1, SSBK2

           J=1

 3         CONTINUE

           NUBK(I)=NUMPT(I)-J+1

           IF (HB(I,J).LE.BKB(I)) THEN

             J=J+1

             GOTO 3

           END IF

           BKT(I)=(HT(I,J)-HT(I,J-1))*(BKB(I)-HB(I,J-1))

     &            /(HB(I,J)-HB(I,J-1))+HT(I,J-1)

        END DO

      END IF

C

C*********************************************************************

C       Output file of the calculated results: OSMB44.CSV
        OPEN(2,FILE='OSMB44.CSV')

        NO=0

        DO ITHETA=1, NTHETA

          THTA=THETA1(ITHETA)*3.14159/180

          DO IKAI=1, NKAI

            DO IFN=1, NFN

              DO IRAM=1, NRA

                NO=NO+1

C

                OMEGA=6.28318/PERI(IRAM,IFN,IKAI)

C

C      *********** Frictional Component ( frictional component) ************

C      Ref. H.Kato (JZK. No.102) and S.Tamiya et al (JZK. No.132)

C

        SF=L*(1.7*D+CB*B)

        RF=((0.887+0.145*CB)*(1.7*D+CB*B)-2*OGD*D)/3.1415

        BFHAT=0.787*SF*RF**2.*SQRT(OMEGA*NUE*B/19.6133)

     &         /(NABLA*B**2.)*(1.+4.1*FUN(IFN)/OMEGA*SQRT(9.80665/L))

C

C

C      ********** Wave Making Component (wave making component) **********
C      Generally, this component should be calculated by the external potential code.
C      Ref. Y.Ikeda et al (JZK. No.143)

C    

        GUZAID=OMEGA**2.*D/9.80665

        A1=1+GUZAID**(-1.2)*EXP(-2*GUZAID)

        A2=0.5+GUZAID**(-1)*EXP(-2*GUZAID)

        LOMEGA=OMEGA*FUN(IFN)*SQRT(L/9.80665)

        SIT=20.*(LOMEGA-0.3)

        TAH=(EXP(SIT)-EXP(-SIT))/(EXP(SIT)+EXP(-SIT))

        BWHAT=BWOSM*.5*(((A2+1)+(A2-1)*TAH)

     &        +(2*A1-A2-1)*EXP(-150*(LOMEGA-.25)**2))

C

C

C     ********** Lift Component **********

C      Ref. Y.Ikeda et al (JZK. No.143)

C

        IF( CM.LE.0.92 )THEN

          KAPA=0.0

        END IF

        IF( (CM.LE.0.97).AND.(CM.GT.0.92) )THEN

          KAPA=0.1

        END IF

        IF( CM.GT.0.97 )THEN

          KAPA=0.3

        END IF

        KN=6.28319*D/L+KAPA*(4.1*B/L-0.045)

        OG=OGD*D

        LO=0.3*D

        LR=0.5*D

        BLHAT=L*D*KN*LO*LR*FUN(IFN)*0.5/(NABLA*B**2)

     &        *SQRT(0.5*L*B)*(1-1.4*OG/LR+0.7*OG**2/(LO*LR))

C

C      ******* Eddy Making Component ********

C             RDP=3: for barge, RDP=1: for conventional round hull
        IF (RDP.EQ.3) THEN

C

C      ******* Eddy Making Component for type barge ships *******

C       Ref. Y.Ikeda  et al(Proc. Ship Motion, Wave Loads and

C                          Propaulsive Performance in a Seaway,1984)

C          Y.Ikeda, T.Fujiwara, T.Katayama (Proc ISOPE Vol.3, 1993)

C

        OG=OGD*D

        DO I=1, N

          BE2(I)=(2/3.141592)*DX(I)**4/NABLA/BX(I)**2*(B/2/9.8)**.5

     &           *(HO(I)**2+1-OG/DX(I))*(HO(I)**2

     &           +(1-OG/DX(I))**2)*THTA*OMEGA

        END DO

        BEHAT=0

        DO I=1,N-1

          XHA=(X(I+1)-X(I))

          SBE=XHA*(BE2(I)+BE2(I+1))/2

          BEHAT=BEHAT+SBE

        END DO

        BEHAT=BEHAT*L/(X(N)-X(1))

        IF (FUN(IFN).EQ.0) then 

          BEHAT=BEHAT

          GOTO 5215

        END IF

        AK=OMEGA/FUN(IFN)*SQRT(L/9.8)

        BEHAT=BEHAT*(.04*AK)**2/((.04*AK)**2+1)

 5215   CONTINUE 

C

C

        GOTO 5064

        END IF

C

C     *********** Eddy Making Component **********

C     Ref. Y.Ikeda et al (JZK. No.142, JZK. No.143)

C     Ref.   T.Katayama et al  (PJASNAOE No.33) 
C

        DO J=1, N

          AHO=HO(J)/(1-OGD)

          SGUMA=(SIG(J)-OGD)/(1-OGD)

          E=(AHO-1)/(AHO+1)

          E2=E**2

          A=4*SGUMA*(1-E2)/3.1415+E2

          O=-A/(A+3)

          O2=SQRT(O**2-(A-1)/(A+3))

          A3=O+O2

          A1=E*(1+A3)

          AM=BX(J)/(1+A1+A3)*0.5

          AA1=A1*(1+A3)/A3*0.25

          IF (AA1.GT.1) THEN

            AA1=1

          END IF

          IF (AA1.LE.-1) THEN

            AA1=-1

          END IF

          DO I=1, 2

            CALL ARC(AA1,ACS)

            LTHETA=.5*ACS

            IF (I.EQ.1) THEN

              LTHETA=0

            END IF

            AH=1+A1**2+9*A3**2+2*A1*(1-3*A3)*COS(2*LTHETA)

     &         -6*A3*COS(4*LTHETA)

            AA=-2*A3*COS(5*LTHETA)+A1*(1-A3)*COS(3*LTHETA)

     &         +((6-3*A1)*A3**2+(A1**2-3*A1)*A3+A1**2)*COS(LTHETA)

            BB=-2*A3*SIN(5*LTHETA)+A1*(1-A3)*SIN(3*LTHETA)

     &         +((6+3*A1)*A3**2+(3*A1+A1**2)*A3+A1**2)*SIN(LTHETA)

            V(I)=2*AM*SQRT(AA**2+BB**2)/AH

            RMAX1(I)=AM*SQRT(((1+A1)*SIN(LTHETA)-A3*SIN(3*LTHETA))**2

     &               +((1-A1)*COS(LTHETA)+A3*COS(3*LTHETA))**2)

          END DO

          RMAX=RMAX1(1)

          VMAX=V(1)

          IF (RMAX1(1).LE.RMAX1(2)) THEN

            GOTO 5159

          END IF

          GOTO 5161

 5159     RMAX=RMAX1(2)

          VMAX=V(2)

 5161     RMEAN=2*DX(J)*(1-OGD)*SQRT(AHO*SGUMA/3.1415)

          P1=VMAX/RMEAN

          P2=RMAX/RMEAN

          PP3=P1+P2

          GAMMA=(1+4*EXP(-165000*(1-SGUMA)**2))*PP3

          CP=0.5*(0.87*EXP(-GAMMA)-4*EXP(-0.187*GAMMA)+3)

          SIT=20*(SIG(J)-0.7)

          CALL TANH(TAH,SIT)

          F1=0.5*(1+TAH)

          F2=0.5*(1-COS(3.1415*SIG(J)))-1.5*(1-EXP(-5*(1-SIG(J))))

     &       *SIN(3.1415*SIG(J))**2

          IF (SIG(J).GT.1) THEN

            SIG(J)=1.

          END IF

          R=2*DX(J)*SQRT(HO(J)*(SIG(J)-1)/(-0.8584))

          RD=R/DX(J)

          IF ((HO(J).LE.1).AND.(RD.GE.AHO)) THEN

            R=.5*BX(J)

          END IF

          IF ((HO(J).GE.1).AND.(RD.GE.1)) THEN

            R=DX(J)

          END IF

          RD=R/DX(J)

          CR1(J)=RMAX**2/DX(J)**2*CP*((1-F1*RD)*(1-3/2*OGD-F1*RD)

     &          +F2*(HO(J)-F1*RD)**2)

          IF( CR1(J).LT.0 )THEN

            CR1(J)=0

          END IF

        END DO

        DO K=1, 21

          X1(K)=0+.5*(K-1)

          MAX=N

          MAXX=SC

          NN=MAX

          M1=1

          M2=0

          XXX=X1(K)

          DO JJ=1, MAXX

            X11(JJ)=X(JJ)

            Y11(JJ)=CR1(JJ)

          END DO

          CALL HOKAN1(NN,IL,XXX,YX,Y2,M1,M2,X11,Y11,WX,WY)

          CR(K)=Y2

          DAM=YX

        END DO

        CR(1)=1.5*(1-3/2*OGD)

        CR(21)=1.5*(1-3/2*OGD)

        SAM=0

        DO K=1, 10

          K2=2*K

          K1=K2-1

          K3=K2+1

          SAM1=CR(K1)+4*CR(K2)+CR(K3)

          SAM=SAM+SAM1

        END DO

        CRT=SAM/60

        BEHAT=4*L*D**4/3/3.1415*OMEGA*SQRT(B/19.6)/NABLA/B**2*CRT*THTA

        DO ISSN=1,N 

          BE2(ISSN)=CR1(ISSN)*4*D**4/3/3.1415*OMEGA

     &              *SQRT(B/19.6)/NABLA/B**2*THTA 

        END DO

C

        IF (FUN(IFN).EQ.0) then 

          BEHAT=BEHAT

          GOTO 5211

        END IF 

C

        AK=OMEGA/FUN(IFN)*SQRT(L/9.8)

        BEHAT=BEHAT*(.04*AK)**2/((.04*AK)**2+1)

        DO ISSN=1,N 

          BE2(ISSN)=BE2(ISSN)*(.04*AK)**2/((.04*AK)**2+1) 

        END DO

 5211   CONTINUE

C

C

C =====Start of calculation of Bilge keel component======

C

 5064   IF (BBK.LE.1E-10) THEN

        GOTO 5067

        END IF

C

C      TYPE_BK=1: simplified method
C      TYPE_BK=2: integration method
C

        IF (TYPE_BK.EQ.1) THEN

C

C      ********** Damping due to Bilge Keels  (The simplified estimation method)**********
C      Ref. Y.Ikeda et al (JZK. No.161 , JZK. No.165)

C

        XBK(1)=XBK1

        XBK(11)=XBK2

        DO I=2, 10

          XBK(I)=XBK(I-1)+(XBK2-XBK1)*.1

        END DO

        MAX=N

        DO I=1, 11

          MAXX=SC 

          NN=MAX  

          M1=1  

          M2=0

          XXX=XBK(I)

          DO JJ=1, MAXX

            X11(JJ)=X(JJ) 

            Y11(JJ)=HO(JJ)

          END DO

          CALL HOKAN1(NN,IL,XXX,YX,Y2,M1,M2,X11,Y11,WX,WY)

          HO1=Y2

          DAM=YX

          MAXX=SC 

          NN=MAX 

          M1=1 

          M2=0

          XXX=XBK(I)

          DO JJ=1, MAXX

            X11(JJ)=X(JJ)

            Y11(JJ)=SIG(JJ)

          END DO

          CALL HOKAN1(NN,IL,XXX,YX,Y2,M1,M2,X11,Y11,WX,WY)

          SG1=Y2

          DAM=YX

          MAXX=SC 

          NN=MAX 

          M1=1 

          M2=0

          XXX=XBK(I)

          DO JJ=1, MAXX

            X11(JJ)=X(JJ) 

            Y11(JJ)=DX(JJ)

          END DO

          CALL HOKAN1(NN,IL,XXX,YX,Y2,M1,M2,X11,Y11,WX,WY)

          DX1=Y2

          DAM=YX

          MAXX=SC 

          NN=MAX  

          XXX=XBK(I)

          M1=1 

          M2=0

          XXX=XBK(I)

          DO JJ=1, MAXX

            X11(JJ)=X(JJ)

            Y11(JJ)=BX(JJ)

          END DO

          CALL HOKAN1(NN,IL,XXX,YX,Y2,M1,M2,X11,Y11,WX,WY)

          BX1=Y2

          DAM=YX

          R=2*DX1*sqrt(HO1*(SG1-1)/(-.8585))

          RD=R/DX1

          IF ((HO1.LE.1).AND.(RD.GE.HO1)) then 

            R=.5*BX1

          END IF

          IF ((HO1.GT.1).AND.(RD.GE.1)) then 

            R=DX1

          END IF 

          RD=R/DX1

          F=1+.3*EXP(-160*(1-SG1))

          RBK=DX1*SQRT((HO1-.2929*RD)**2+(1-OGD-.2929*RD)**2)

          M1=RD

          M2=OGD

          M3=1-M1-M2

          M4=HO1-M1

          M5=(.414*HO1+.0651*M1**2-(.382*HO1+.0106)*M1)

     &       /((HO1-.215*M1)*(1-.215*M1))

          M6=(.414*HO1+.0651*M1**2-(.382+.0106*HO1)*M1)

     &       /((HO1-.215*M1)*(1-.215*M1))

          SO=.3*(3.1415*F*RBK*THTA)+1.95*BBK

          M7=SO/DX1-.25*3.1415*M1

          R1=.25*3.1415*R

          IF (SO.LT.R1) then 

             M7=0

          END IF

          M8=M7+.414*M1

          IF (SO.LT.R1) then 

             M8=M7+1.414*(1-cos(SO/R))*M1

          END IF

          A=(M3+M4)*M8-M7**2

          BB=M4**3/3/(HO1-.215*M1)+(1-M1)**2*(2*M3-M2)

     &       /6/(1-.215*M1)+M1*(M3*M5+M4*M6)

          CPPLAS=1.2

          CPMINS=-22.5*BBK/(3.1415*RBK*F*THTA)-1.2

          CD=CPPLAS-CPMINS

C
C      *** BBKhat for unit length ***

C

          RATIO(I)=RBK*BBK*CD/(RBK*BBK*CD+.5*DX1**2

     &           *(-A*CPMINS+BB*CPPLAS))

          BBKHAT(I)=8*RBK**2*OMEGA*SQRT(B/19.6)

     &              *THTA*F**2/(3*3.1415*NABLA*B**2)

     &              *(RBK*BBK*CD+.5*DX1**2*(-A*CPMINS+BB*CPPLAS))

        END DO

C

C      *** BBKhat for Three Dimentional Ship Form ***

C

        SAM=0

        DO I=1,5

          I2=2*I

          SAM1=BBKHAT(I2-1)+4*BBKHAT(I2)+BBKHAT(I2+1)

          SAM=SAM+SAM1

        END DO

        BKHAT=SAM*(XBK2-XBK1)*.1/3*L*.1

C

C

               ELSE

C

C      ********** Damping due to Bilge Keels **********

C      Pressure integration along a hull surface
C      Ref. Y.Ikeda, T.Katayama et al (JKSNAJ. No.222)

C
         BBK0=0

         BBK1=0

         BBK2=0

         BBK3=0

         RO=102

         KG=(1-OGD)*D

         DO P=SSBK1, SSBK2

           SGUMA=(SIG(P)-OGD)/(1-OGD)

C      ******* normal pressure component *******

C      ******* F1, R, CD----->FBN

           K=NUBK(P)

           XX0=XX(P,K)

           XX1=BKB(P)

           XX2=XX(P,K+1)

           YY0=YY(P,K)

           YY1=BKT(P)

           YY2=YY(P,K+1)

             IF (XX2.EQ.XX1) THEN 

               XX2=XX(P,K+2) 

               LOW=102

             END IF 

             YY2=YY(P,K+2)

             F1=1+.3*EXP((-1)*160*(1-SGUMA))

             R=SQRT((KG-YY1)**2+XX1**2)

             CD=22.5*BBK/(3.14159*F1*R*THTA)+2.4

             FBN=8/3*RO*R**2*BBK*THTA*F1**2*CD*OMEGA/6.28

C       ******* XA,YA ***** Gradient of hull surface.  Quadratic function approximation
             IF (XX0.EQ.XX2) THEN 

               XA=0   

               YA=YY1 

               GOTO 3550

             END IF

             IF (YY0.EQ.YY2) THEN 

               XA=XX1 

               YA=KG

               GOTO 3550

             END IF

             FD0=YY0/((XX0-XX1)*(XX0-XX2))

             FD1=YY1/((XX1-XX0)*(XX1-XX2))

             FD2=YY2/((XX2-XX0)*(XX2-XX1))

             FD=2*XX1*(FD0+FD1+FD2)-(XX1+XX2)

     &          *FD0-(XX0+XX2)*FD1-(XX0+XX1)*FD2

             XA=(XX1+FD*(YY1-KG))/(1+FD**2)

             YA=(FD*XX1+FD**2*(YY1-KG)+KG)/(1+FD**2)

C        ******* LB1,BN

 3550        LB1=SQRT((XA-XX1)**2+(YA-YY1)**2)

             BN=FBN*LB1*2

C

C        ******* negative pressure component *******

C        ******* S0,CPN----->FBSN

             S0=.3*3.14159*F1*R*THTA+1.95*BBK

             CPN=1.2-CD

             FBSN=8/3*RO*R**2*F1**2*THTA*CPN*OMEGA/6.28

C        ******* starboard side

C        ******* XX0,YY0,XX1,YY1

             MPN=0 

             LB2S0=0 

             OWARI=0 

             BSN=0 

             BSNS=0 

             BSNP=0

             K=NUBK(P)

             XX0=BKB(P)  

             YY0=BKT(P)  

             XX1=XX(P,K+1) 

             YY1=YY(P,K+1) 

             GOTO 3950

 3900        K=K+1

             XX0=XX(P,K) 

             YY0=YY(P,K) 

             XX1=XX(P,K+1) 

             YY1=YY(P,K+1) 

             GOTO 3950

 3920        XX1=(XX1-XX0)*(S0+LB2-LB2S0)/LB2+XX0

             YY1=(YY1-YY0)*(S0+LB2-LB2S0)/LB2+YY0

C         ******* XM,YM,XA,YA

 3950        XM=(XX0+XX1)/2 

             YM=(YY0+YY1)/2

             IF (XX0.EQ.XX1) THEN 

               XA=0

               YA=YM

               GOTO 4010

             END IF

             IF (YY0.EQ.YY1) THEN 

               XA=XM

               YA=KG

               GOTO 4010

             END IF

             XA=(YM+(XX0-XX1)/(YY0-YY1)*XM-KG)*(XX0-XX1)

     &          *(YY0-YY1)/((YY0-YY1)**2+(XX0-XX1)**2)

             YA=(YY0-YY1)/(XX0-XX1)*XA+KG

C         ******* LB2,LB3----->BSNS

 4010        LB2=SQRT((XX0-XX1)**2+(YY0-YY1)**2)

             LB2S0=LB2S0+LB2

             IF ((OWARI.EQ.0).AND.(S0.LT.LB2S0)) THEN

               OWARI=1 

               GOTO 3920

             END IF

             LB3=SQRT((XM-XA)**2+(YM-YA)**2)

             BSNS=BSNS+FBSN*LB2/LB3*(XA*(YM-YA)-(YA-KG)*(XM-XA))

             IF (OWARI.EQ.1) THEN

               OWARI=0

               GOTO 4110

             END IF

             GOTO 3900

C         ******* port side

C         ******* XX0,YY0,XX1,YY1

 4110        LB2S0=0

             K=2*NUMPT(P)-1-NUBK(P)

             XX0=-BKB(P) 

             YY0=BKT(P)   

             XX1=XX(P,K+1) 

             YY1=YY(P,K+1)

             GOTO 4160

 4140        K=K+1

             XX0=XX(P,K) 

             YY0=YY(P,K)  

             XX1=XX(P,K+1) 

             YY1=YY(P,K+1)

 4160        IF (YY1.GT.D) THEN 

               OWARI=1

               XX1=(XX1-XX0)*(D-YY0)/(YY1-YY0)+XX0

               YY1=D

             END IF

             GOTO 4210

 4180        XX1=(XX1-XX0)*(S0+LB2-LB2S0)/LB2+XX0

             YY1=(YY1-YY0)*(S0+LB2-LB2S0)/LB2+YY0

C         ******* XM,YM,XA,YA

 4210         XM=(XX0+XX1)/2 

             YM=(YY0+YY1)/2

             IF (XX0.EQ.XX1) THEN 

               XA=0  

               YA=YM 

               GOTO 4270            

             END IF

             IF (YY0.EQ.YY1) THEN 

               XA=XM

               YA=KG

               GOTO 4270

             END IF

             XA=(YM+(XX0-XX1)/(YY0-YY1)*XM-KG)*(XX0-XX1)

     &          *(YY0-YY1)/((YY0-YY1)**2+(XX0-XX1)**2)

             YA=(YY0-YY1)/(XX0-XX1)*XA+KG

C         ******* LB2,LB3----->BSNP

 4270        LB2=SQRT((XX0-XX1)**2+(YY0-YY1)**2)

             LB2S0=LB2S0+LB2

             IF ((OWARI.EQ.0).AND.(S0.LT.LB2S0)) THEN 

               OWARI=1 

               GOTO 4180

             END IF

             LB3=SQRT((XM-XA)**2+(YM-YA)**2)

             BSNP=BSNP+FBSN*LB2/LB3*(XA*(YM-YA)-(YA-KG)*(XM-XA))

             IF (OWARI.EQ.1) THEN 

               OWARI=0 

               GOTO 4350

             END IF

           GOTO 4140

 4350      BSN=BSNS+BSNP

C

C      ******* positive pressure component *******

C      ******* CPP

           CPP=1.2

C      ******* starboard side

C      ******* GLWBK

           GLWBK=0 

           GLBKK0=0

           GLBKK1=0 

           BSP=0 

           BSPS=0 

           BSPP=0

           DO K=1, NUBK(P)

             XX0=XX(P,K) 

             YY0=YY(P,K) 

             XX1=XX(P,K+1) 

             YY1=YY(P,K+1)

             IF (YY1.GE.D) THEN 

               goto 4520

             END IF

             IF ((YY1.LT.D).AND.(YY0.GE.D)) then 

               XX0=(XX1-XX0)*(D-YY0)/(YY1-YY0)+XX0 

               YY0=D

             END IF

C             

             IF (K.EQ.NUBK(P)) THEN

               XX1=BKB(P)

               YY1=BKT(P)

             END IF

             GLWBK=GLWBK+SQRT((XX0-XX1)**2+(YY0-YY1)**2)

 4520      END DO

C      ******* GLBKK0, GLBKK1, FBSP0, FBSP1

           DO K=1, NUBK(P)

             XX0=XX(P,K)

             YY0=YY(P,K)

             XX1=XX(P,K+1)

             YY1=YY(P,K+1)

             IF (YY1.GE.D) then 

               goto 4740

             END IF

             IF ((YY1.LT.D).AND.(YY0.GE.D)) then 

               XX0=(XX1-XX0)*(D-YY0)/(YY1-YY0)+XX0

               YY0=D

             END IF

             IF (K.EQ.NUBK(P)) THEN

               XX1=BKB(P)    

               YY1=BKT(P)

             END IF

             GLBKK0=GLBKK1

             GLBKK1=GLBKK0+SQRT((XX0-XX1)**2+(YY0-YY1)**2)

             FBSP0=8/3*RO*R**2*THTA*F1**2*CPP*OMEGA/6.28*GLBKK0/GLWBK

             FBSP1=8/3*RO*R**2*THTA*F1**2*CPP*OMEGA/6.28*GLBKK1/GLWBK

C      ******* XM, YM, XA, YA, LB2, LB3----->BSP

             XM=(FBSP0+2*FBSP1)/(3*FBSP0+3*FBSP1)*(XX1-XX0)+XX0

             YM=(FBSP0+2*FBSP1)/(3*FBSP0+3*FBSP1)*(YY1-YY0)+YY0

             IF (XX0.EQ.XX1) THEN 

               XA=0 

               YA=YM

               GOTO 4700

             END IF 

             IF (YY0.EQ.YY1) THEN 

               XA=XM

               YA=KG

               GOTO 4700

             END IF

             XA=(YM+(XX0-XX1)/(YY0-YY1)*XM-KG)*(XX0-XX1)

     &          *(YY0-YY1)/((YY0-YY1)**2+(XX0-XX1)**2)

             YA=(YY0-YY1)/(XX0-XX1)*XA+KG

 4700        LB2=SQRT((XX0-XX1)**2+(YY0-YY1)**2)

             LB3=SQRT((XM-XA)**2+(YM-YA)**2)

             BSPS=BSPS+(FBSP0+FBSP1)/2*LB2/LB3

     &            *(XA*(YM-YA)-(YA-KG)*(XM-XA))

 4740      END DO

C       ******* port side

C       ******* GLWBK

           GLWBK=0 

           GLBKK0=0 

           GLBKK1=0 

           OWARI=0

           KSTA=NUMPT(P)     

           KEND=2*NUMPT(P)-1-NUBK(P)

           DO K=KSTA, KEND

             XX0=XX(P,K)

             YY0=YY(P,K)

             XX1=XX(P,K+1)

             YY1=YY(P,K+1)

             IF (K.EQ.KEND) THEN  

               XX1=-BKB(P)

               YY1=BKT(P)

             END IF

             GLWBK=GLWBK+SQRT((XX0-XX1)**2+(YY0-YY1)**2)

           END DO

C       ******* GLBKK0, GLBKK1, FBSP0, FBSP1

           DO K=KSTA, KEND

             XX0=XX(P,K)

             YY0=YY(P,K)

             XX1=XX(P,K+1)

             YY1=YY(P,K+1)

             IF (K.EQ.KEND) THEN

               XX1=-BKB(P)   

               YY1=BKT(P)

             END IF

             GLBKK0=GLBKK1

             GLBKK1=GLBKK0+SQRT((XX0-XX1)**2+(YY0-YY1)**2)

             FBSP0=8/3*RO*R**2*THTA*F1**2*CPP*OMEGA/6.28*GLBKK0/GLWBK

             FBSP1=8/3*RO*R**2*THTA*F1**2*CPP*OMEGA/6.28*GLBKK1/GLWBK

C       ******* XM, YM, XA, YA, LB2, LB3----->BSP

             XM=(FBSP0+2*FBSP1)/(3*FBSP0+3*FBSP1)*(XX1-XX0)+XX0

             YM=(FBSP0+2*FBSP1)/(3*FBSP0+3*FBSP1)*(YY1-YY0)+YY0

             IF (XX0.EQ.XX1) THEN 

               XA=0  

               YA=YM 

               GOTO 4990

             END IF

             IF (YY0.EQ.YY1) THEN 

               XA=XM 

               YA=KG 

               GOTO 4990

             END IF

             XA=(YM+(XX0-XX1)/(YY0-YY1)*XM-KG)*(XX0-XX1)

     &          *(YY0-YY1)/((YY0-YY1)**2+(XX0-XX1)**2)

             YA=(YY0-YY1)/(XX0-XX1)*XA+KG

 4990        LB2=SQRT((XX0-XX1)**2+(YY0-YY1)**2)

             LB3=SQRT((XM-XA)**2+(YM-YA)**2)

             BSPP=BSPP+(FBSP0+FBSP1)/2*LB2/LB3

     &             *(XA*(YM-YA)-(YA-KG)*(XM-XA))

           END DO

           BSP=BSPS+BSPP

C

C********************************************************************

           BBK0=BBK1

           BBK1=BN+BSN+BSP

           BBK2=BBK0+BBK1

           IF (P.EQ.SSBK1) then 

             BBK3=BBK3+L*(CSD(P)-XBK1)/10*BBK1

             GOTO 5140

           END IF

           BBK3=BBK3+L*(CSD(P)-CSD(P-1))/10*BBK2/2

 5140      IF (P.EQ.SSBK2) THEN 

             BBKDAMP=BBK3+L*(XBK2-CSD(P))/10*BBK1

           END IF

C       *******

         END DO

         BKHAT=BBKDAMP/(RO*NABLA*B**2)*SQRT(B/19.6)

C       *******

C=======================================

                END IF

                B2KHAT=BKHAT

 5067           IF( BBK.LE.1E-10)THEN

                  B2KHAT=0.

                END IF

C =====Here, the program about calculation of a bilge-keel component finishes.======

C      Below, a calculated result is written out to a file.
                BB44H=BFHAT+BWHAT+BLHAT+BEHAT+B2KHAT

                B44   =BB44H*(RO*NABLA*B**2)/SQRT(B/19.6)

                BF    =BFHAT *(RO*NABLA*B**2)/SQRT(B/19.6)

                BW   =BWHAT *(RO*NABLA*B**2)/SQRT(B/19.6)

                BL    =BLHAT *(RO*NABLA*B**2)/SQRT(B/19.6)

                BE    =BEHAT *(RO*NABLA*B**2)/SQRT(B/19.6)

                BBBK =B2KHAT*(RO*NABLA*B**2)/SQRT(B/19.6)

C=========== The write-in line to the file of a roll damping calculated result.  Below, it is shown in order of writing.
C               roll amp.[deg]、Fn、wave length: λ[m]、wave direction: χ[deg]

                WRITE(2,100) THETA1(ITHETA), FUN(IFN), RAM(IRAM)

     &                  ,KAI(IKAI), BB44H, BFHAT, BWHAT, BLHAT, BEHAT

     &                  , B2KHAT, B44, BF, BW, BL, BE, BBBK

  100           FORMAT(16(F20.9,','))

C============

                END DO

              END DO

            END DO

          END DO

        END DO

        CLOSE(2)

        STOP

        END

C

C

C

C       ************   Subroutine   *************

C      ********** Lagrange 3 Points Interpolation **********

      SUBROUTINE HOKAN1(NN,IL,XXX,YX,Y2,M1,M2,X11,Y11,WX,WY)

      IMPLICIT DOUBLE PRECISION (A-H,O-Z) 

      DOUBLE PRECISION NUE,L,KAI,L0,L1,L2,NABLA,KW,KAI1,LTHETA,MGA,KN

     &    ,LO,LR,KG,LB1,LB2,LB3,LB2S0,M1,M2,M3,M4,M5,M6,M7,M8

      INTEGER BWOSM,TYPE_BK,RDP,CSDMIN,SSBK1,SSBK2

      PARAMETER SC=200,PT=200,NNRA=70,NNFN=20,NNKAI=35,NNRD=30

C

      DIMENSION  X11(SC),Y11(SC),WX(SC),WY(SC)

C       

C

         N1=NN-1

         DO IL=2,N1

           IF(XXX.LE.X11(IL)) THEN

             GOTO 5324

           END IF

         END DO

 5324    I1=IL-1

         IF (XXX.GT.X11(N1)) THEN

           I1=NN-2

         END IF

         I2=I1+2

         DO IL=I1,I2

           II=IL+1-I1

           WX(II)=X11(IL)

           WY(II)=Y11(IL)

         END DO

         IF (M1.NE.1) THEN

           GOTO 5334

         END IF

           CALL LAG3(XXX,Y2,X11,Y11,WX,WY)

 5334     IF (M2.NE.1) THEN

           GOTO 5336

         END IF

         YX=0

 5336  RETURN

       END

C

      SUBROUTINE LAG3(XXX,Y2,X11,Y11,WX,WY)

      IMPLICIT DOUBLE PRECISION (A-H,O-Z) 

      PARAMETER SC=200,PT=200,NNRA=70,NNFN=20,NNKAI=35,NNRD=30

      DOUBLE PRECISION NUE,L,KAI,L0,L1,L2,NABLA,KW,KAI1,LTHETA,MGA,KN

     &    ,LO,LR,KG,LB1,LB2,LB3,LB2S0,M1,M2,M3,M4,M5,M6,M7,M8

      INTEGER BWOSM,TYPE_BK,RDP,CSDMIN,SSBK1,SSBK2

      DIMENSION  X11(SC),Y11(SC),WX(SC),WY(SC)

C       

         Y2=0

         DO I3=1, 3

           W=1

           Z=1

           DO J3=1, 3

             IF(J3.EQ.I3) THEN

               GOTO 5350

             END IF

             W=W*(XXX-WX(J3))

             Z=Z*(WX(I3)-WX(J3))

 5350      END DO

           Y2=Y2+WY(I3)*W/Z

         END DO

       RETURN

       END

C

C

C      ********** arc cos **********

      SUBROUTINE ARC(AA1,ACS)

      IMPLICIT DOUBLE PRECISION (A-H,O-Z) 

C

         IF(ABS(AA1).EQ.1) THEN 

           ACS=0 

           GOTO 5361

         END IF

         ACS=-ATAN(AA1/SQRT(-AA1*AA1+1))+3.14159/2

 5361  RETURN

       END

C

C

C      ********** hyperbolic tangent **********

      SUBROUTINE TANH(TAH,SIT)

      IMPLICIT DOUBLE PRECISION (A-H,O-Z) 

C

         TAH=(EXP(SIT)-EXP(-SIT))/(EXP(SIT)+EXP(-SIT))

       RETURN

       END

C       

C
END of Document
� EMBED Equation.3  ���

